Prob. 19.21 - Cantilevered beam

The stress σ_x has no dependency on the material properties, and is not influenced by material viscoelasticity.
The correspondence-principle recipe starts by putting the deflection relation in the Laplace plane (C is the compliance operator):

```
> v_bar:=(x^2*(3*L-x)/(6*Ix))*C*F_bar;
```

$$v\_bar := \frac{1}{6}\frac{x^2\,(3\,L-x)\,C\,F\_bar}{Ix}$$

Transform of load:

```
> with(inttrans):F_bar:=laplace(F*Heaviside(t),t,s);
```

$$F\_bar := \frac{F}{s}$$

Compliance operator:

```
> C:=Cg+Cv/(tau*(s+1/tau));
```

$$C := Cg + \frac{Cv}{\tau\left(s+\frac{1}{\tau}\right)}$$

Invert to get deflection in time plane:

```
> v(t):=invlaplace(v_bar,s,t);
```

$$\mathrm{v}(t) := \frac{1}{6}\frac{x^2\,(3\,L-x)\,F\left(-Cv\,\mathbf{e}^{\left(-\frac{t}{\tau}\right)}+Cg+Cv\right)}{Ix}$$

Simplifying manually to standard form:

$$v(x,t) = \frac{x^2\,(3L-x)}{6I}\cdot F\cdot\left[C_g + C_v\left(1-e^{-t/\tau}\right)\right]$$

Superposition approach: write load and compliance as time functions:

```
> F:= (t)-> F[0]*Heaviside(t);
```

$$F := t \to F_0\,\mathrm{Heaviside}(t)$$

```
> C[crp]:= (t)-> C[g]+C[v]*(1-exp(-t/tau));
```

$$C_{crp} := t \to C_g + C_v\left(1-\mathbf{e}^{\left(-\frac{t}{\tau}\right)}\right)$$

Superposition integral:

```
> v(t):=(x^2*(3*L-x)/(6*Ix))*int(C[crp](t-xi)*diff(F(xi),xi),xi=-inf
  inity..t);
```

$$v(t) := -\frac{1}{6} \frac{x^2 (3L - x) F_0 \text{Heaviside}(t) \left(-C_g - C_v + C_v \mathbf{e}^{\left(-\frac{t}{\tau}\right)}\right)}{Ix}$$

This can be reduced to the same form obtained previously.
Examine deflection function for arbitrary choice of parameters:

```
> plot(subs({L=2,x=1,F[0]=1,C[g]=1,C[v]=1,tau=1,Ix=1},v(t)),t=-5..5)
;
```



Prob. 19.22 - Rigid die

Define Poisson (N) and tensile modulus (EE) operators in terms of dilatation (K) and shear (G) operators:

```
> N:=(3*K-2*G)/(6*K+2*G);EE:=(9*G*K)/(3*K+G);
```

$$N := \frac{3K - 2G}{6K + 2G}$$

$$EE := 9 \frac{G K}{3K + G}$$

SLS expressions for G and K:

```
> G:=Gr+((Gg-Gr)*s)/(s+(1/tau_G));
```

$$G := Gr + \frac{(Gg - Gr)\, s}{s + \frac{1}{tau\_G}}$$

```
> K:=Kr+((Kg-Kr)*s)/(s+(1/tau_K));
```

$$K := Kr + \frac{(Kg - Kr)\, s}{s + \frac{1}{tau\_K}}$$

Pick model parameters from Fig. 17. Relaxation times (tau_K and tau_G) are those times at which the relaxation has dropped (1/e) of its total value.

```
> Digits:=20:Gg:=8.8*10^8:Gr:=2.4*10^5:tau_G:=.001:
> Kg:=6.2*10^9:Kr:=1.7*10^9:tau_K:=.0005:
> sigybar:=sigma[y]/s;
```

$$sigybar := \frac{\sigma_y}{s}$$

Transverse stress:

```
> sigxbar:=N*sigybar/(1-N):
> sigma[y]:=10*10^6;
```

$$\sigma_y := 10000000$$

```
> sigma[x]:=invlaplace(sigxbar,s,t):
> with(plots):semilogplot(sigma[x],t=10^(-5)..10,labels=[t,`sigma[x]
  `],numpoints=5000);
```



Note that the transverse stress becomes equal to the vertical stress (i.e. the stress state becomes hydrostatic) as the relaxation completes and the material approaches a rubbery state.

Vertical strain:

```
> epsybar:=(1+N)*(1-2*N)*sigybar/(EE*(1-N)):
> epsilon[y]:=invlaplace(epsybar,s,t):
> loglogplot(epsilon[y],t=10^(-5)..10,labels=[t,`epsilon[y]`],numpoi
  nts=5000);
```



We might expect the (1-2*nu) factor to drive the strain to zero as nu approaches 0.5, but the tensile modulus is also dropping substantially, and the strain is observed to rise. The material does not become fully rubbery, and maintains a finite compressibility.